Minimi Quadrati

Enrico Bertolazzi

Introduzione

Scopo: dati i punti $p_0, p_1, ..., p_N$ dove

$$p_i = \begin{bmatrix} x_i \\ \\ y_i \end{bmatrix}$$

trovare il polinomio $P(x) = \sum_{i=0}^{M} a_i x^i$ che meglio approssima i punti dati secondo i minimi quadrati

Soluzione: basta risolvere il sistema $A\ a = b$ dove

$$A_{i,j} = \sum_{k=0}^{N} x_k^{(i+j-2)} \qquad b_i = \sum_{k=0}^{N} x_k^{(i-1)} y_k$$

Carica le librerie

```
> initialize ;
  with(LinearAlgebra):
  with(plots):
```

initialize

```
Warning, the name changecoords has been redefined
```

Definisce la procedura minq

```
> minq := proc(M,xy)
    local P, i, j, k, N, x, X, Y, A, b, a, Aij, bi ;

    N   := nops(xy) ;
    X   := Vector([seq(xy[i][1],i=1..N)]) ;
    Y   := Vector([seq(xy[i][2],i=1..N)]) ;

    Aij := (i,j) -> add(X[k]^(i+j-2),k=1..N) ;
    bi  := (i)   -> add(Y[k]*X[k]^(i-1),k=1..N) ;

    A   := Matrix(M+1,Aij) ;
    b   := Vector(M+1,bi) ;
```

```
    ## risolve il sistema
    a   := LinearSolve(A,b) ;
    P   := unapply(add(a[k+1]*x^k,k=0..M),x);

    return P, A, b, a ;
  end :
```

Procedura di Stampa

```
> minq_print := proc(P, xy)
    local i, N, X, Y, xmin, xmax, dx, scarto, GA, GB ;

    N := nops(xy) ;
    X := Vector([seq(xy[i][1],i=1..N)]) ;
    Y := Vector([seq(xy[i][2],i=1..N)]) ;

    xmin   := min(seq(X[i],i=1..N)) ;
    xmax   := max(seq(X[i],i=1..N)) ;
    dx     := (xmax - xmin) / 20 ;
    scarto := add((P(X[i])-Y[i])^2, i=1..N)/N ;

    print (Transpose(Matrix(xy))) ;

    ## risultati
    print("Scarto^2 = ",scarto) ;
    print("Poly     = ",P) ;

    GA := plot(P,xmin-dx..xmax+dx,style=line,thickness=2,color=blue):
    GB := plot(xy,style=point,symbol=circle,symbolsize=25,color=red):

    display({GA,GB},axes=boxed,title=`Minimi Quadrati`);

  end :
```

Esempio d'uso

```
> # definisce la funzione da approssimare
  f := x -> x/(1+x*x) ;
```

$$f := x \to \frac{x}{1 + x\,x}$$

```
> # definisce i punti da interpolare
  pnts:=[seq([k,f(k)], k=-4..4)] ;
```

$$pnts := \left[\left[-4, \frac{-4}{17} \right], \left[-3, \frac{-3}{10} \right], \left[-2, \frac{-2}{5} \right], \left[-1, \frac{-1}{2} \right], [0, 0], \left[1, \frac{1}{2} \right], \left[2, \frac{2}{5} \right], \left[3, \frac{3}{10} \right], \left[4, \frac{4}{17} \right] \right]$$

```
> # risolve il problema e stampa i risultati
```

```
P, A, b, a := minq(3,pnts) ;
minq_print(P, pnts) ;
```

$$P, A, b, a := P, \begin{bmatrix} 9 & 0 & 60 & 0 \\ 0 & 60 & 0 & 708 \\ 60 & 0 & 708 & 0 \\ 0 & 708 & 0 & 9780 \end{bmatrix}, \begin{bmatrix} 0 \\ \frac{534}{85} \\ 0 \\ \frac{4566}{85} \end{bmatrix}, \begin{bmatrix} 0 \\ \frac{2303}{8415} \\ 0 \\ \frac{-241}{16830} \end{bmatrix}$$

$$\begin{bmatrix} -4 & -3 & -2 & -1 & 0 & 1 & 2 & 3 & 4 \\ \frac{-4}{17} & \frac{-3}{10} & \frac{-2}{5} & \frac{-1}{2} & 0 & \frac{1}{2} & \frac{2}{5} & \frac{3}{10} & \frac{4}{17} \end{bmatrix}$$

$$\text{"Scarto^2 = ", } \frac{12764}{715275}$$

$$\text{"Poly} \quad = \text{", } x \rightarrow \frac{2303}{8415} x - \frac{241}{16830} x^3$$

Minimi Quadrati



Soluzione con le primitive Maple

```
> with(stats):
```

```
> Xvalues := [seq(pnts[k][1], k=1..nops(pnts))] ;
  Yvalues := [seq(pnts[k][2], k=1..nops(pnts))] ;
```

$$Xvalues := [-4, -3, -2, -1, 0, 1, 2, 3, 4]$$

$$Yvalues := \left[\frac{-4}{17}, \frac{-3}{10}, \frac{-2}{5}, \frac{-1}{2}, 0, \frac{1}{2}, \frac{2}{5}, \frac{3}{10}, \frac{4}{17} \right]$$

```
> # cancella le variabili a,b,c,d se eventualmente inizializzate
  a := 'a' : b := 'b' : c := 'c' : d := 'd' :
```

```
> fit[leastsquare[[x,y], y=a+b*x+c*x^2+d*x^3,
  {a,b,c,d}]]([Xvalues,Yvalues]);
```

$$y = \frac{2303}{8415} x - \frac{241}{16830} x^3$$

```
>
```